

Component Based Modeling for Relativistic Electrons

Sebastian Micluța-Câmpeanu^{1,2}

¹JuliaHub

²University of Bucharest

Abstract

ElectronDynamicsModels.jl provides composable models for relativistic electron dynamics within ModelingToolkit, enabling radiation reaction and photon emission studies. Building on LaserTypes.jl, it offers AD compatibility and symbolic manipulation through modular components connected via MTK.

Introduction

ElectronDynamicsModels.jl provides a framework for modeling relativistic electron dynamics within ModelingToolkit. Building upon our previous work in LaserTypes.jl, this new approach offers:

- Automatic differentiation compatibility
- Symbolic manipulation capabilities
- Composable model architecture
- Swappable radiation models

Physical Models

Covariant Formulation

Equations of Motion

$$\frac{dx^\mu}{d\tau} = u^\mu$$

$$m \frac{du^\mu}{d\tau} = F_{\text{total}}^\mu = F_{\text{Lorentz}}^\mu + F_{\text{rad}}^\mu$$

$$\text{where } F_{\text{Lorentz}}^\mu = qF^{\mu\nu}u_\nu$$

Modular Code Structure

```
@component function ParticleDynamics(; name, mass = 1.0, spacetime)
    @unpack c = spacetime

    @variables begin
        t(τ), [description = "Universal time"]
        γ(τ), [description = "Lorentz factor"]
        x(τ)[1:4], [guess = [c * t, 0, 0, 0]]
        u(τ)[1:4], [guess = [c, 0, 0, 0]]
        p(τ)[1:4]
        F_total(τ)[1:4]
    end
    @parameters m = mass

    eqs = [
        t ~ x[1] / c
        p ~ m * u
        u[1] ~ γ * c
        dτ(x) ~ u
        dτ(u) ~ F_total / m
    ]

    System(eqs, τ, [t, γ, x, u, p, F_total], [m];
        name, systems = [spacetime])
end
```

External Fields

Plane Wave

Monochromatic plane wave

$$\mathbf{E} = E_0 \hat{x} \cos(\mathbf{k} \cdot \mathbf{r} - \omega t)$$

$$\mathbf{B} = \frac{\mathbf{k} \times \mathbf{E}}{\omega} = (\hat{k} \times \hat{x}) \frac{E_0}{c} \cos(\mathbf{k} \cdot \mathbf{r} - \omega t)$$

Propagating along \hat{k} with linear polarization in \hat{x} direction

Gaussian Laser Pulse

Electric field at focus

$$E(r, z, t) = E_0 \frac{w_0}{w(z)} \exp\left(-\frac{r^2}{w(z)^2}\right) \exp(i(kz - \omega t + \varphi(r, z)))$$

with beam waist w_0 , Rayleigh range $z_R = \pi \frac{w_0^2}{\lambda}$




Figure 1: 3D visualization of focused Gaussian beam

Creating Custom Components

```
# Define custom field configurations
@component function CustomField(; name, spacetime)
    @named field_dynamics = EMFieldDynamics(; spacetime)
    @variables x(τ)[1:4] t(τ)
    E, B = field_dynamics.E, field_dynamics.B

    # Define your field components
    eqs = [
        E[1] ~ my_Ex_function(x, t),
        E[2] ~ my_Ey_function(x, t),
        E[3] ~ my_Ez_function(x, t),
        B[1] ~ my_Bx_function(x, t),
        B[2] ~ my_By_function(x, t),
        B[3] ~ my_Bz_function(x, t)
    ]

    System(eqs, τ, name, systems=[field_dynamics])
end

# Compose with electron
@named custom_field = CustomField(; spacetime)
@named electron = ChargedParticle(external_field = custom_field)
```

ModelingToolkit Architecture

External Fields

PlaneWave, GaussLaser

Faraday tensor: $F^{\mu\nu} = \begin{pmatrix} 0 & -E_x/c & -E_y/c & -E_z/c \\ E_x/c & 0 & -B_z & B_y \\ E_y/c & B_z & 0 & -B_x \\ E_z/c & -B_y & B_x & 0 \end{pmatrix}$

Radiation Models

Abraham-Lorentz, Landau-Lifshitz

Abraham-Lorentz: $F_{\text{rad}}^\mu \propto \frac{d^2 u^\mu}{d\tau^2}$

ParticleDynamics

$$dx^\mu/d\tau = u^\mu$$

$$mdu^\mu/d\tau = F_{\text{total}}^\mu$$

$$u_\mu u^\mu = -c^2$$

Composed System: ChargedParticle

$$F_{\text{Lorentz}}^\mu = qF^{\mu\nu}u_\nu$$

$$F_{\text{total}}^\mu = F_{\text{Lorentz}}^\mu + F_{\text{rad}}^\mu$$

Figure 2: Component-based architecture with swappable models

```
# Modular force composition
# Without radiation:
F_total ~ F_lorentz

# Lorentz force from field tensor
F_lorentz[μ] ~ q * sum(F[μν][ν, v] * u[v] for v in 1:4)

# With radiation reaction (optional):
if radiation_model == :abraham_lorentz
    @named radiation = AbrahamLorentzRadiation(
        charge = q,
        spacetime = spacetime,
        particle = particle
    )
    push!(systems, radiation)
    F_total ~ F_lorentz + radiation.F_rad
end
```

System Transformation

Unsimplified System (24 equations)

Differential equations (2):

$$D(\text{particle.x}) = \text{particle.u}$$

$$D(\text{particle.u}) = \frac{\text{particle.F}_{\text{total}}}{\text{particle.m}}$$

Algebraic equations (22):

$$x = \text{particle.x}$$

$$u = \text{particle.u}$$

$$\text{plane_wave.x} = \text{particle.x}$$

$$\text{plane_wave.t} = \text{particle.t}$$

$$\text{particle.t} = \frac{\text{particle.x}_1}{c}$$

$$\text{particle.p} = \text{particle.m} \times \text{particle.u}$$

$$\text{particle.u}_1 = c \times \text{particle.}\gamma$$

$$E_1 = A \cos(\mathbf{k} \cdot \mathbf{x} - \omega t)$$

$$E_2 = 0$$

$$E_3 = 0$$

$$B_1 = 0$$

$$B_2 = A \frac{\cos(\mathbf{k} \cdot \mathbf{x} - \omega t)}{c}$$

$$B_3 = 0$$

$$\lambda = 2\pi \frac{c}{\omega}$$

$$F_{\text{lorentz}} = qF^{\mu\nu}(g_{\mu\nu}u)$$

$$\text{particle.F}_{\text{total}} = F_{\text{lorentz}}$$

$$\text{Field tensor: } F^{\mu\nu} = \begin{pmatrix} 0 & -\frac{E_1}{c} & -\frac{E_2}{c} & -\frac{E_3}{c} \\ \frac{E_1}{c} & 0 & -B_3 & B_2 \\ \frac{E_2}{c} & B_3 & 0 & -B_1 \\ \frac{E_3}{c} & -B_2 & B_1 & 0 \end{pmatrix}$$

$$\text{Stress-energy tensor: } T^{\mu\nu} = \frac{1}{\mu_0} [F^{\mu\alpha} F_{\alpha}^{\nu} - \frac{1}{4} g^{\mu\nu} F_{\alpha\beta} F^{\alpha\beta}]$$

Compiled System (8 ODEs only!)

Full equations after structural simplification:

$$D(x^1) = u^1$$

$$D(x^2) = u^2$$

$$D(x^3) = u^3$$

$$D(x^4) = u^4$$

$$D(u^1) = (-g_{21}u^1 - g_{22}u^2 - g_{23}u^3 - g_{24}u^4) \frac{Aq \cos(\varphi)}{cm}$$

$$D(u^2) = \frac{qA \cos(\varphi)}{m} [g_{11}u^1 + g_{12}u^2 + g_{13}u^3 + g_{14}u^4 + g_{11}u^1 + g_{12}u^2 + g_{13}u^3 + g_{14}u^4]$$

$$D(u^3) = 0$$

$$D(u^4) = (-g_{21}u^1 - g_{22}u^2 - g_{23}u^3 - g_{24}u^4) \frac{Aq \cos(\varphi)}{cm}$$

$$\text{where } \varphi = -\omega \frac{x^1}{c} + k_1 x^2 + k_2 x^3 + k_3 x^4$$

Get Started

GitHub: github.com/SebastianM-C/ElectronDynamicsModels.jl

Poster created with assistance from Claude

Figure-8 Motion Example

Charged particles in plane waves can exhibit figure-8 motion [1], providing an excellent numerical accuracy test.

```
using ElectronDynamicsModels, ModelingToolkit
using OrdinaryDiffEqVern9, OrdinaryDiffEqTsit5

# Physical parameters
k = 1; c = 1; a0 = 1.0

# Setup plane wave and electron
@named plane_wave = PlaneWave(
    amplitude = a0,
    frequency = c * k,
    k_vector = [0, 0, k]
)
@named electron = ChargedParticle(
    external_field = plane_wave,
    radiation_model = nothing
)
sys = mtkcompile(electron)

# Initial conditions for figure-8 orbit
v0_z = -c / ((2/a0)^2 + 1)
γ0 = 1.0 / sqrt(1 - v0_z^2)
u0 = [
    sys.x => [0.0, 0.0, 0.0, 0.0],
    sys.u => [γ0, 0.0, 0.0, γ0*v0_z]
]

# Solve with different accuracies
prob = ODEProblem(sys, u0, (0.0, 500.0))
sol_tsit5 = solve(prob, Tsit5())
sol_vern9 = solve(prob, Vern9(),
    abstol=1e-9, reltol=1e-9)
```

Numerical Results

Conservation Metrics

Four-velocity norm: $u_\mu u^\mu = -c^2$ (constant)

Lightfront momentum: $p^- = p^0 - p^3$ (conserved in plane wave)

Conservation of these quantities validates numerical accuracy

Solver Comparison




Figure 3: Figure-8 trajectories with conservation analysis

Metric	Tsit5 (1e-6)	Vern9 (1e-6)	Vern9 (1e-9)
Time steps	761	366	1287
Function evaluations	4563	5842	20578
Accepted steps	760	365	1286
4-velocity norm error	3.65×10^{-4}	1.65×10^{-7}	7×10^{-12}

Energy Evolution




Figure 4: Energy oscillations in plane wave interaction

In plane wave interactions, energy is **not conserved** due to work done by the electromagnetic field. The electron gains and loses energy periodically as it oscillates in the wave field.

References

Bibliography

[1] E. S. Sarachik and G. T. Schappert, "Classical Theory of the Scattering of Intense Laser Radiation by Free Electrons," *Physical Review D*, vol. 1, no. 10, pp. 2738–2753, 1970, doi: 10.1103/PhysRevD.1.2738.